
Area of a Plane Region

Objective In this lab you will use the definite integral to determine the area of two-
dimensional regions.

Background In principle every area can be computed using either horizontal or vertical slicing.
However, in some cases one approach will be simpler to set up or the resulting
integrals will be simpler to evaluate. The different formulations are explored
in more detail in Examples 3, 5, and 6. The relevant functions and points are
entered and found in Example 1. Example 2 contains the plot commands needed
to create the image at the bottom of this page. Example 4 shows how inverse
functions can be found, that is, convert from y = f(x) to x = g(y).

Example 2 introduces the scaling=constrained optional argument for the
plot command and the textplot command for placing text within a plot.

Discussion Enter, and execute, the following Maple commands in a Maple worksheet.

Example 1: Problem Setup

> restart; # clear Maple’s memory
> with( plots ); # load package
> f1 := x -> sqrt(x/2); # define function
> f2 := x -> 1/sqrt(9*x+1); # define function
> P1 := plot( [ f1(x), f2(x) ], x=0..2 ): # plot functions
> P1; # display basic plot
> x1 := solve( f1(x)=f2(x), x ); # x-coord of intersection
> q1 := f1(x1) - f2(x1); # is this 0?
> simplify( q1 ); # complete check
> y1 := f1(x1); # y-coord of intersection
> y2 := f2(2); # y-coord at right edge

Example 2: Plot Production

> P2 := implicitplot( { y=1, x=2 }, x=0..2, # plot bounding rectangle
> y=0..1, color=blue ):

> P3 := textplot( # label four regions
> [[0.1,0.5,"A"],[0.6,0.8,"B"],

> [1.3,0.5,"C"],[0.6,0.2,"D"]] ):

> P4 := textplot( [x1,y1+0.1,"(x1,y1)"], # label intersection
> align=ABOVE ):

> P5 := textplot( [2,y2,"(2,y2)"], # label pt on right edge
> align=RIGHT ):

> P6 := textplot([[1,f1(1)-0.1,"y=f1(x)"], # label functions
> [1,f2(1)-0.1,"y=f2(x)"]] ):

> display( [P1, P2, P3, P4, P5, P6], # display figure
> view=[0..2.5,0..1.25],

> scaling=constrained );

Example 3: Area of Region A by Vertical Slices

> q1 := Int( f2(x)-f1(x), x=0..x1 ); # area as integral
> q2 := value( q1 ); # exact area
> evalf( q2 ); # approximate value

Example 4: Find Inverse Functions

> r1 := solve( y=f1(x), x ); # express x as fn of y

> g1 := unapply( r1, y ); # inverse function
> r2 := solve( y=f2(x), x ); # express x as fn of y

> g2 := unapply( r2, y ); # inverse function

Example 5: Area of Region A by Horizontal Slices

> s1 := Int( g1(y)-0, y=0..y1 ); # lower portion of region A
> s2 := Int( g2(y)-0, y=y1..1 ); # upper portion of region A
> s3 := s1 + s2; # area as integral
> s4 := value( s3 ); # exact area
> evalf( s4 ); # approximate value
> simplify( q2 - s4 ); # agree with Example 3?

Example 6: Area of a Circle

> t1 := x^2 + y^2 = R^2; # equation of circle
> t2 := solve( t1, y ); # finds two solutions
> top := t2[1]; # top of circle
> bottom := t2[2]; # bottom of circle
> t3 := Int( top-bottom, x=-R..R ); # definite integral for area
> value( t3 ); # what’s wrong here?
> value( t3 ) assuming R>0; # try again

Notes

(1) The scaling=constrained optional argument in a plot (or display) command instructs
Maple to create the plot using the same scale for both axes.

(2) The figure was saved to a file by selecting the Export As entry on the context menu that
appears when the right mouse button is pressed when the cursor is on the plot.

(3) The assuming clause can be added to many Maple commands to provide additional in-
formation about a problem. Without this information Maple assumes all variables are
complex-valued.

Questions

(1) Set up a single integral for the area of region B. What is the area of region B?
(2) Set up a single integral for the area of region C. What is the area of region C?
(3) Express the area of region D as the sum of two integrals. What is the area of region D?
(4) Set up an integral for the area of the ellipse

x2

r2
+

y2

R2
= 1.

What is the value of this integral? What is special about the ellipse in the case when R = r?
What value does your formula for the area give in the case when R = r?

